
Grammar of The μPascal Programming Language

Notation:
( ) grouping
| alternatives ("or")
[ ] 0 or 1 occurrences
{} 0 or more occurrences
italic nonterminals (pojmovi)
bold terminals (simboli)

Patterns

digit

→ "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" 
letter

→  "_" | "a" | "A" | "b" | "B" | "c" | "C" | "d" | "D" | "e" | "E" | "f"
 | "F" | "g" | "G" | "h" | "H" | "i" | "I" | "j" | "J" | "k" | "K" | "l"
 | "L" | "m" | "M" | "n" | "N" | "o" | "O" | "p" | "P" | "q" | "Q" | "r"
 | "R" | "s" | "S" | "t" | "T" | "u" | "U" | "v" | "V" | "w" | "W" | "x"
 | "X" | "y" | "Y" | "z" | "Z" 

constant

→ digit +
identifier

→ letter ( letter | digit ) *

Productions

program

→ "program" identifier ";" [ "var" variable_list ] [ procedure_list ]
"begin" statement_list "end"

variable_list

→ variable { ";" variable }

variable

→ identifier ":" "int"

procedure_list

→ procedure { ";" procedure }

procedure

→ "procedure" identifier [ "(" variable_list ")" ] [ "var" variable_list ] 
  "begin" statement_list "end"

statement_list

→ [ statement { ";" statement } ]

statement

→ assignment_or_procedure_call_statement
| if_statement
| while_statement
| "begin" statement_list "end"

assignment_or_procedure_call_statement

→ identifier ( ":=" expression | [ "(" expression { "," expression } ")" ] )

expression

→ ["+"|"-"] operand { ("*"|"/") operand } 
{ ("+"|"-") operand { ("*"|"/") operand } }

operand

→ constant
| identifier
| "(" expression ")"

if_statement

→ "if" condition "then" statement [ "else" statement ]

condition

→ expression ( "<" | ">" | "<=" | ">=" | "=" | "<>" ) expression

while_statement

→ "while" condition "do" statement

Comment: { a comment }

Space and newline are symbol separators.
Symbol Token

eof _EOF
"," _COMMA
";" _SEMICOLON
"(" _LEFT_PAR
")" _RIGTH_PAR
"+" _PLUS
"-" _MINUS
"*" _MULTIPLY
"/" _DIVIDE
":=" _ASSIGN
":" _COLON
"=" _EQ
"<=" _LE
"<>" _NE
"<" _LT
">=" _GE
">" _GT
letter ( letter | digit ) * _IDENTIFIER
"program" _PROGRAM
"var" _VAR
"proc" _PROC
"begin" _BEGIN
"end" _END
"if" _IF
"then" _THEN
"else" _ELSE
"while" _WHILE
"do" _DO
"int" _INT
digit + _CONSTANT

Operator Precedence

All operators on the same line have the same precedence. The first line has the highest precedence. 

() the highest precedence
+  - unary
*  /
+  - arithmetic
<  <=  >=  >=  =  <>
:= the lowest precedence

